Hydrogen atom according to wave mechanics in cartesian coordinates

J. F. Ogilvie, Centre for Experimental and Constructive Mathematics, Department of Mathematics, Simon Fraser University, Burnaby, British Columbia V5A 1S6 Canada

```
> restart:
```

In section 12b.53a we demonstrate that a direct solution of Schroedinger's temporally dependent or independent equation is impractical in cartesian coordinates x, y, z because the coulonbic potential energy proportional to $\frac{1}{r}$ or $\frac{1}{\sqrt{x^2+y^2+z^2}}$ prevents the separation of those variables. G. R. Fowles (*American Journal of Physics* 30 (4) 308, 1962) sought to prove that, with amplitude function $\psi(x, y, z)$ as an extended function $f(x, y, z, r)$ of those four variables and with $r^2 = x^2 + y^2 + z^2$, a partial separation of variables was practicable. The eigenfunctions for angular momentum are derived with no reference to polar angles. We remind that the surfaces of constant x, y, z are planes perpendicular to the corresponding axes of coordinates.

```
> plots[implicitplot3d]([x=0,y=0,z=0], x=-2..2, y=-2..2, z=-2..2,

      title="planes in cartesian coordinates -- x=0 red, y=0
  blue, z=0 green",
      axes=boxed,
  colour=[red,blue,green],titlefont=[TIMES,BOLD,14]);
```

planes in cartesian coordinates -- x=0 red, y=0 blue, z=0 green



\>

To develop the lhamiltonian for this system we proceed as follows to form the required second derivatives. As

```
> r := sqrt(x^2 + y^2 +z^2);
```

$$r := \sqrt{x^2 + y^2 + z^2}$$

```
> Diff(r,x) = diff(r,x);
```

$$\frac{\partial}{\partial x}(\sqrt{x^2 + y^2 + z^2}) = \frac{x}{\sqrt{x^2 + y^2 + z^2}}$$

so $\frac{\partial}{\partial x} r = \frac{x}{r}$

```
> psi := f(x,y,z,r);
```

$$\psi := f(x, y, z, \sqrt{x^2+y^2+z^2})$$

```
> Diff('psi',x) = convert(diff(psi,x), diff);
```

$$\frac{\partial}{\partial x}\psi = \left(\frac{\partial}{\partial x} f(x, y, z, t1)\right)\Bigg|_{t1=\sqrt{x^2+y^2+z^2}} + \frac{\left(\frac{\partial}{\partial t1} f(x, y, z, t1)\right)\Big|_{t1=\sqrt{x^2+y^2+z^2}} x}{\sqrt{x^2+y^2+z^2}}$$

which we rewrite as

```
> Diff('psi',x) = diff(f(x,y,z,'r'),x) +
  x*diff(f(x,y,z,'r'),'r')/'r';   # **
```

$$\frac{\partial}{\partial x}\psi = \left(\frac{\partial}{\partial x} f(x, y, z, r)\right) + \frac{x\left(\frac{\partial}{\partial r} f(x, y, z, r)\right)}{r}$$

For the second derivative,

```
> Diff('psi',x$2) = expand(convert(diff(psi,x$2), diff));
```

$$\frac{\partial^2}{\partial x^2}\psi = \left(\frac{\partial^2}{\partial x^2} f(x, y, z, t1)\right)\Bigg|_{t1=\sqrt{x^2+y^2+z^2}} + \frac{2\left(\frac{\partial^2}{\partial x\,\partial t1} f(x, y, z, t1)\right)\Big|_{t1=\sqrt{x^2+y^2+z^2}} x}{\sqrt{x^2+y^2+z^2}}$$
$$+ \frac{x^2\left(\frac{\partial^2}{\partial t1^2} f(x, y, z, t1)\right)\Big|_{t1=\sqrt{x^2+y^2+z^2}}}{x^2+y^2+z^2} - \frac{\left(\frac{\partial}{\partial t1} f(x, y, z, t1)\right)\Big|_{t1=\sqrt{x^2+y^2+z^2}} x^2}{(x^2+y^2+z^2)^{(3/2)}}$$
$$+ \frac{\left(\frac{\partial}{\partial t1} f(x, y, z, t1)\right)\Big|_{t1=\sqrt{x^2+y^2+z^2}}}{\sqrt{x^2+y^2+z^2}}$$

which we rewrite as

```
> dx2 := Diff('psi',x$2) = Diff(f(x,y,z,'r'),x$2) +
  2*x*Diff(f(x,y,z,'r'),x,'r')/'r'
                + x^2*Diff(f(x,y,z,'r'),'r','r')/'r'^2 +
  Diff(f(x,y,z,'r'),'r')/'r'
                - x^2*Diff(f(x,y,z,'r'),'r')/'r'^3;
```

$$dx2 := \frac{\partial^2}{\partial x^2}\psi = \left(\frac{\partial^2}{\partial x^2} f(x, y, z, r)\right) + \frac{2\,x\left(\frac{\partial^2}{\partial r\,\partial x} f(x, y, z, r)\right)}{r} + \frac{x^2\left(\frac{\partial^2}{\partial r^2} f(x, y, z, r)\right)}{r^2}$$
$$+ \frac{\frac{\partial}{\partial r} f(x, y, z, r)}{r} - \frac{x^2\left(\frac{\partial}{\partial r} f(x, y, z, r)\right)}{r^3}$$

with analogous expressions for $\frac{\partial^2}{\partial y^2}\psi$ and $\frac{\partial^2}{\partial z^2}\psi$.

```
> dy2 := Diff('psi',y$2) = Diff(f(x,y,z,'r'),y$2) +
  2*y*Diff(f(x,y,z,'r'),y,'r')/'r'
                    + y^2*Diff(f(x,y,z,'r'),'r','r')/'r'^2 +
  Diff(f(x,y,z,'r'),'r')/'r'
                    - y^2*Diff(f(x,y,z,'r'),'r')/'r'^3;
```

$$dy2 := \frac{\partial^2}{\partial y^2}\psi = \left(\frac{\partial^2}{\partial y^2}\mathrm{f}(x, y, z, r)\right) + \frac{2\,y\left(\frac{\partial^2}{\partial r\,\partial y}\mathrm{f}(x, y, z, r)\right)}{r} + \frac{y^2\left(\frac{\partial^2}{\partial r^2}\mathrm{f}(x, y, z, r)\right)}{r^2} + \frac{\frac{\partial}{\partial r}\mathrm{f}(x, y, z, r)}{r} - \frac{y^2\left(\frac{\partial}{\partial r}\mathrm{f}(x, y, z, r)\right)}{r^3}$$

```
> dz2 := Diff('psi',z$2) = Diff(f(x,y,z,'r'),z$2) +
  2*z*Diff(f(x,y,z,'r'),z,'r')/'r'
                    + z^2*Diff(f(x,y,z,'r'),'r','r')/'r'^2 +
  Diff(f(x,y,z,'r'),'r')/'r'
                    - z^2*Diff(f(x,y,z,'r'),'r')/'r'^3;
```

$$dz2 := \frac{\partial^2}{\partial z^2}\psi = \left(\frac{\partial^2}{\partial z^2}\mathrm{f}(x, y, z, r)\right) + \frac{2\,z\left(\frac{\partial^2}{\partial r\,\partial z}\mathrm{f}(x, y, z, r)\right)}{r} + \frac{z^2\left(\frac{\partial^2}{\partial r^2}\mathrm{f}(x, y, z, r)\right)}{r^2} + \frac{\frac{\partial}{\partial r}\mathrm{f}(x, y, z, r)}{r} - \frac{z^2\left(\frac{\partial}{\partial r}\mathrm{f}(x, y, z, r)\right)}{r^3}$$

To form the laplacian operator as the sum of the three second derivatives, we add the right sides of the above three equations and collect the terms to obtain

```
> Diff('psi',x$2) + Diff('psi',y$2) + Diff('psi',z$2) =
  Diff(f(x,y,z,'r'),'x','x')
                          + Diff(f(x,y,z,'r'),'y','y') +
  Diff(f(x,y,z,'r'),'z','z')
                          + (2/'r')*(x*Diff(f(x,y,z,'r'),'x','r')
  + y*Diff(f(x,y,z,'r'),'y','r')
                          + z*Diff(f(x,y,z,'r'),'z','r')) +
  3*Diff(f(x,y,z,'r'),'r')/'r'
                          - Diff(f(x,y,z,'r'),'r')/'r';
```

$$\left(\frac{\partial^2}{\partial x^2}\psi\right) + \left(\frac{\partial^2}{\partial y^2}\psi\right) + \left(\frac{\partial^2}{\partial z^2}\psi\right) = \left(\frac{\partial^2}{\partial x^2}\mathrm{f}(x, y, z, r)\right) + \left(\frac{\partial^2}{\partial y^2}\mathrm{f}(x, y, z, r)\right) + \left(\frac{\partial^2}{\partial z^2}\mathrm{f}(x, y, z, r)\right) + \frac{2\left(x\left(\frac{\partial^2}{\partial r\,\partial x}\mathrm{f}(x, y, z, r)\right) + y\left(\frac{\partial^2}{\partial r\,\partial y}\mathrm{f}(x, y, z, r)\right) + z\left(\frac{\partial^2}{\partial r\,\partial z}\mathrm{f}(x, y, z, r)\right)\right)}{r}$$

$$+\frac{2\left(\frac{\partial}{\partial r}\mathrm{f}(x, y, z, r)\right)}{r}$$

For the hydrogen atom as a central field for which potential energy V depends on only r, we assume that the part of $\mathrm{f}(x, y, z, r)$ dependent on r is separable, so that $\mathrm{f}(x, y, z, r) = \mathrm{F}(x, y, z)\,\mathrm{R}(r)$. The laplacian operator becomes accordingly,

```
> R('r')*(Diff(F(x,y,z),x$2) + Diff(F(x,y,z),y$2) +
  Diff(F(x,y,z),z$2))
    + (2/'r')*Diff(R('r'),'r')*(x*Diff(F(x,y,z),x) +
  y*Diff(F(x,y,z),x) + z*Diff(F(x,y,z),x))
    + F(x,y,z)*Diff(R('r'),'r','r') +
  2*F(x,y,z)*Diff(R('r'),'r')/'r';
```

$$\mathrm{R}(r)\left(\left(\frac{\partial^2}{\partial x^2}\mathrm{F}(x, y, z)\right)+\left(\frac{\partial^2}{\partial y^2}\mathrm{F}(x, y, z)\right)+\left(\frac{\partial^2}{\partial z^2}\mathrm{F}(x, y, z)\right)\right)$$
$$+\frac{2\left(\frac{d}{dr}\mathrm{R}(r)\right)\left(x\left(\frac{\partial}{\partial x}\mathrm{F}(x, y, z)\right)+y\left(\frac{\partial}{\partial x}\mathrm{F}(x, y, z)\right)+z\left(\frac{\partial}{\partial x}\mathrm{F}(x, y, z)\right)\right)}{r}$$
$$+\mathrm{F}(x, y, z)\left(\frac{d^2}{dr^2}\mathrm{R}(r)\right)+\frac{2\,\mathrm{F}(x, y, z)\left(\frac{d}{dr}\mathrm{R}(r)\right)}{r}$$

and Schroedinger's temporally independent equation, after division through by $\mathrm{F}(x, y, z)$, becomes

```
> R('r')*(Diff(F(x,y,z),x$2) + Diff(F(x,y,z),y$2) +
  Diff(F(x,y,z),z$2))
    + (2/'r')*Diff(R('r'),'r')*(x*Diff(F(x,y,z),x) +
  y*Diff(F(x,y,z),x) + z*Diff(F(x,y,z),x))
    + F(x,y,z)*Diff(R('r'),'r','r') +
  2*F(x,y,z)*Diff(R('r'),'r')/'r' +
  8*Pi^2*mu/h^2*(E-V('r'))*R('r')=0;
```

$$\mathrm{R}(r)\left(\left(\frac{\partial^2}{\partial x^2}\mathrm{F}(x, y, z)\right)+\left(\frac{\partial^2}{\partial y^2}\mathrm{F}(x, y, z)\right)+\left(\frac{\partial^2}{\partial z^2}\mathrm{F}(x, y, z)\right)\right)$$
$$+\frac{2\left(\frac{d}{dr}\mathrm{R}(r)\right)\left(x\left(\frac{\partial}{\partial x}\mathrm{F}(x, y, z)\right)+y\left(\frac{\partial}{\partial x}\mathrm{F}(x, y, z)\right)+z\left(\frac{\partial}{\partial x}\mathrm{F}(x, y, z)\right)\right)}{r}$$
$$+\mathrm{F}(x, y, z)\left(\frac{d^2}{dr^2}\mathrm{R}(r)\right)+\frac{2\,\mathrm{F}(x, y, z)\left(\frac{d}{dr}\mathrm{R}(r)\right)}{r}+\frac{8\,\pi^2\,\mu\,(E-\mathrm{V}(r))\,\mathrm{R}(r)}{h^2}=0$$

We seek solutions for which $\mathrm{F}(x, y, z)$ satisfies Laplace's equation in three spatial dimensions.

$$\left(\frac{\partial^2}{\partial x^2}\mathrm{F}(x, y, z)\right)+\left(\frac{\partial^2}{\partial y^2}\mathrm{F}(x, y, z)\right)+\left(\frac{\partial^2}{\partial z^2}\mathrm{F}(x, y, z)\right) = 0$$

In section 7.401 we show that $F_l(x, y, z) = (a\,x + b\,y + c\,z)^l$ is a solution of Laplace's equation providing that $a^2 + b^2 + c^2 = 0$, which implies that at least one coefficient a, b, c must be complex, and that l must be an integer for F_l to be singly valued. In section 7.401 we derive further that

$$x\left(\frac{\partial}{\partial x}F_l(x, y, z)\right)+y\left(\frac{\partial}{\partial x}F_l(x, y, z)\right)+z\left(\frac{\partial}{\partial x}F_l(x, y, z)\right) = l\,F_l(x, y, z)$$

The Schroedinger equation above consequently reduces to this radial equation, containing reduced mass μ for the system,

```
> Diff(R('r'),'r','r')+2*(l+1)/'r'*Diff(R('r'),'r')+
  8*Pi^2*mu/h^2*(E-V('r'))*R('r') = 0;
```

$$\left(\frac{d^2}{dr^2}\mathrm{R}(r)\right)+\frac{2\,(l+1)\left(\frac{d}{dr}\mathrm{R}(r)\right)}{r}+\frac{8\,\pi^2\,\mu\,(E-\mathrm{V}(r))\,\mathrm{R}(r)}{h^2}=0$$

For an atom of atomic number Z with one electron in SI units, $\mathrm{V}(r) = -\frac{Z\,e^2}{4\,\pi\,\varepsilon_0\,r}$. Following Fowles, we introduce dimensionless variable $\rho = \alpha\,r$ and parameter $\lambda = \frac{Z\,e^2\sqrt{\mu}}{2\,\varepsilon_0\,h\sqrt{-2\,E}}$ for eigenvalues in which $\alpha = \frac{8\,\pi^2\,\mu\,Z\,e^2}{h^2\,\varepsilon_0\,\lambda}$, setting $\mathrm{R}(\rho) = \mathbf{e}^{\left(-\frac{\rho}{2}\right)}\,\mathrm{L}(\rho)$, the preceding equation becomes

```
> rho*Diff(L(rho),rho$2)+(2*(l+1)-rho)*Diff(L(rho),rho)+(lambda-l
  -1)*L(rho)=0;
```

$$\rho\left(\frac{d^2}{d\rho^2}\mathrm{L}(\rho)\right)+(2\,l+2-\rho)\left(\frac{d}{d\rho}\mathrm{L}(\rho)\right)+(\lambda-l-1)\,\mathrm{L}(\rho)=0$$

This equation has well behaved solutions in the form of associated Laguerre polynomials $L_k^{(2\,l+1)}(\rho)$, with $\lambda = n = k + l + 1$, a positive integer. The eigenvalues E thus become given according to $E_n = -\frac{2\,\pi^2\,Z^2\,\mu\,e^4}{h^2\,n^2}$, as derived in section 12b.53. The corresponding eigenfunctions are expressed as

```
> psi[n,l](x,y,z,'r')  := F[l](x,y,z)*R[n,l]('r');
```

$$\psi_{n,\,l}(x, y, z, r) := F_l(x, y, z)\,R_{n,\,l}(r)$$

expanded into this formula containing only cartesian coordinates as variables, with several parameters or constants including the effective Bohr radius $a_\mu = \frac{h^2\,\lambda}{8\,\pi^2\,Z\,\mu\,e^2}$ with $a_\mu = \frac{m_e\,a_0}{\mu}$. As $R_{n,\,l}(r)$ is essentially the same here as in spherical polar coordinates, we replace α with a_μ and include the normalizing factor for that radial part from section 12b.53a.

```
> psi[k,l](x,y,z)  :=
  (1/a[mu]/k!*(k+2*l+2)!*(k+2*l+2))^(1/2)/((k+l+1)!*(2+2*l)!)*(a*
  x+b*y+c*z)^l

  *exp(-sqrt(x^2+y^2+z^2)/(a[mu]*(k+l+1)))*LaguerreL(k,2*l+1,sqrt
  (x^2+y^2+z^2)/(a[mu]*(k+l+1)))/((-1)^(2*l+1));
```

$$\psi_{k,l}(x,y,z) := \sqrt{\frac{(k+2\,l+2)!\,(k+2\,l+2)}{a_\mu\,k!}}\,(a\,x+b\,y+c\,z)^l\,\mathbf{e}^{\left(-\frac{\sqrt{x^2+y^2+z^2}}{a_\mu\,(k+l+1)}\right)}\,\mathrm{LaguerreL}\left(k,\,2\,l+1,\,\frac{\sqrt{x^2+y^2+z^2}}{a_\mu\,(k+l+1)}\right)\Big/\left((k+l+1)!\,(2\,l+2)!\,(-1)^{(2\,l+1)}\right)$$

The preceding development that yielded associated Laguerre polynomials in terms of arguments k and l ensured that amplitude function $\psi(x, yu, z, \sqrt{x^2+y^2+z^2})$ has the property of a zero value as $r \to \infty$ so that k and l become quantum numbers taking values of non-negative integers. For the purpose of defining a third quantum number we proceed to consider angular momentum M; from the standard properties of operators for angular momentum in wave mechanics, we have, with $I = \sqrt{-1}$,

```
> M[z]*psi = -I*(h/(2*Pi))*(x*Diff(psi,y)-y*Diff(psi,x));
```

$$M_z\,\psi = \frac{\frac{-1}{2}\,I\,h\left(x\left(\frac{\partial}{\partial y}\psi\right)-y\left(\frac{\partial}{\partial x}\psi\right)\right)}{\pi}$$

```
> M[x]*psi = -I*h*(y*Diff(psi,z)-z*Diff(psi,y))/(2*Pi);
```

$$M_x\,\psi = \frac{\frac{-1}{2}\,I\,h\left(y\left(\frac{\partial}{\partial z}\psi\right)-z\left(\frac{\partial}{\partial y}\psi\right)\right)}{\pi}$$

```
> M[y]*psi = -I*h*(z*Diff(psi,x)-x*Diff(psi,z))/(2*Pi);
```

$$M_y\,\psi = \frac{\frac{-1}{2}\,I\,h\left(z\left(\frac{\partial}{\partial x}\psi\right)-x\left(\frac{\partial}{\partial z}\psi\right)\right)}{\pi}$$

Taking $\psi_{n,l}(x, y, z, r) := F_l(x, y, z)\,R_{n,l}(r)$ as above and making use of the derivative relation marked with ** above, we obtain

```
> M[z]*F[l](x,y,z)  =
  -I*(h/(2*Pi))*(x*Diff(F[l](x,y,z),y)-y*Diff(F[l](x,y,z),y));
```

$$M_z\,F_l(x,y,z) = \frac{\frac{-1}{2}\,I\,h\left(x\left(\frac{\partial}{\partial y}F_l(x,y,z)\right)-y\left(\frac{\partial}{\partial y}F_l(x,y,z)\right)\right)}{\pi}$$

from which $\mathrm{R}(r)$ has cancelled out because $\mathrm{R}(r)$ commutes with angular momentum M. As $\mathrm{R}(r)$ commutes also with $M^2 = {M_x}^2 + {M_y}^2 + {M_z}^2$, from the above formulae we obtain, subject to

$a^2 + b^2 + c^2 = 0$,

```
> M^2*F[l](x,y,z) =
  -(h/(2*Pi))^2*((x*Diff(F[l](x,y,z),y)-y*Diff(F[l](x,y,z),x))^2
                                             +
   (y*Diff(F[l](x,y,z),z)-z*Diff(F[l](x,y,z),y))^2
                                             +
   (z*Diff(F[l](x,y,z),x)-x*Diff(F[l](x,y,z),z))^2);
```

$$M^2\, F_l(x, y, z) = -\frac{1}{4}\, h^2 \left(\left(x\left(\frac{\partial}{\partial y} F_l(x, y, z)\right) - y\left(\frac{\partial}{\partial x} F_l(x, y, z)\right)\right)^2 + \left(y\left(\frac{\partial}{\partial z} F_l(x, y, z)\right) - z\left(\frac{\partial}{\partial y} F_l(x, y, z)\right)\right)^2 + \left(z\left(\frac{\partial}{\partial x} F_l(x, y, z)\right) - x\left(\frac{\partial}{\partial z} F_l(x, y, z)\right)\right)^2\right) \Big/ \pi^2$$

which becomes

```
> M^2*F[l](x,y,z) = (h/(2*Pi))^2*l*(l+1)*F[l](x,y,z);
```

$$M^2\, F_l(x, y, z) = \frac{1}{4}\, \frac{h^2\, l\,(l+1)\, F_l(x, y, z)}{\pi^2}$$

so that the eigenvalues of M^2 are $\left(\frac{h}{2\,\pi}\right)^2 l\,(l+1)$, with no preferred axis. From the formula for $M_z\, \psi$ above, we obtain

```
> M[z]*F[l](x,y,z) =
  -I*(h/(2*Pi))*l*((b*x-a*y)/(a*x+b*y+c*z))*F[l](x,y,z);
```

$$M_z\, F_l(x, y, z) = \frac{\frac{-1}{2}\, I\, h\, l\,(-a\, y + b\, x)\, F_l(x, y, z)}{\pi\,(a\, x + b\, y + c\, z)}$$

which implies that $F_l(x, y, z)$ lacks a definite value of M_z. When we consider that $a^2 + b^2 + c^2 = 0$, this condition is satisfied on defining a, b, c in terms of two arbitrary complex numbers u,v as follows.

```
> a = u^2-v^2;
  b = -I*(u^2+v^2);
  c = -2*u*v;
```

$$a = u^2 - v^2$$

$$b = -I\,(u^2 + v^2)$$

$$c = -2\, u\, v$$

With these definitions,

```
> F[l](x,y,z) = ((u^2-v^2)*x -I*(u^2+v^2)*y -2*u*v*z)^l;
```

$$F_l(x, y, z) = ((u^2 - v^2)\, x - (u^2 + v^2)\, y\, I - 2\, u\, v\, z)^l$$

```
> lhs(%) = expand(rhs(%));
```

$$F_l(x, y, z) = (u^2\,x - v^2\,x - y\,u^2\,I - y\,v^2\,I - 2\,u\,v\,z)^l$$

which on rearrangement becomes

```
> F[l](x,y,z) = (u^2*(x-I*y)-v^2*(x+I*y) - 2*u*v*z)^l;
```

$$F_l(x, y, z) = (u^2\,(x - y\,I) - v^2\,(x + y\,I) - 2\,u\,v\,z)^l$$

Expressed in this way, which is intended to produce amplitude functions of which the plots of their surfaces at selected values of ψ maximally resemble the corresponding surfaces of $\psi_{k,\,l,\,m}(r, \theta, \phi)$ in spherical polar coordinates, $F_l(x, y, z)$ is an homogeneous polynomial of degree $2\,l$ in u and v that contains $2\,l + 1$ terms; their coefficients are polynomials Q of degree l in x, y, z:

```
> F[l](x,y,z) = Sum(u^(l-m)*v^(l+m)*Q[l,m](x,y,z), m=-l..l);
```

$$F_l(x, y, z) = \sum_{m=-l}^{l} u^{(l-m)}\,v^{(l+m)}\,Q_{l,\,m}(x, y, z)$$

For instance, for $l = 0$ we have

```
> F[0](x,y,z) = (u^2*(x-I*y)-v^2*(x+I*y) - 2*u*v*z)^0;
```

$$F_0(x, y, z) = 1$$

which is a trivial case,

```
> Q[0,0](x,y,z) := 1;
```

$$Q_{0,\,0}(x, y, z) := 1$$

leaving only the radial part of the amplitude function. For $l = 1$,

```
> F[1](x,y,z) = (u^2*(x-I*y)-v^2*(x+I*y) - 2*u*v*z)^1;
```

$$F_1(x, y, z) = u^2\,(x - y\,I) - v^2\,(x + y\,I) - 2\,u\,v\,z$$

```
> F[1](x,y,z) := collect(expand((u^2*(x-I*y)-v^2*(x+I*y) -
  2*u*v*z)^1), [u,v]);
```

$$F_1(x, y, z) := u^2\,(x - y\,I) - 2\,u\,v\,z + (-y\,I - x)\,v^2$$

the coefficient of $u\,v$ is real but the coefficients of u^2 and v^2 are complex.

```
> Q[1,-1](x,y,z) := factor(coeff(F[1](x,y,z), u^2));
  Q[1,0](x,y,z) := factor(coeff(F[1](x,y,z)/u, v));
  Q[1,1](x,y,z) := factor(coeff(F[1](x,y,z), v^2));
```

$$Q_{1,\,-1}(x, y, z) := x - y\,I$$

$$Q_{1,\,0}(x, y, z) := -2\,z$$

$$Q_{1,\,1}(x, y, z) := -y\,I - x$$

For $l = 2$, we have

```
> F[2](x,y,z) = (u^2*(x-I*y)-v^2*(x+I*y) - 2*u*v*z)^2;
```

$$F_2(x, y, z) = (u^2\,(x - y\,I) - v^2\,(x + y\,I) - 2\,u\,v\,z)^2$$

```
> F[2](x,y,z) := collect(expand(rhs(%)), [u,v]);
```

$$F_2(x, y, z) := (-2\,I\,x\,y + x^2 - y^2)\,u^4 + (4\,I\,y\,z - 4\,x\,z)\,v\,u^3 + (-2\,x^2 - 2\,y^2 + 4\,z^2)\,v^2\,u^2$$

$$+ (4 I y z + 4 x z) v^3 u + (2 I x y + x^2 - y^2) v^4$$

according to which

```
> Q[2,-2](x,y,z) := factor(coeff(F[2](x,y,z), u^4));
  Q[2,-1](x,y,z) := factor(coeff(F[2](x,y,z)/v, u^3));
  Q[2,0](x,y,z) := factor(coeff(F[2](x,y,z)/v^2, u^2));
  Q[2,1](x,y,z) := factor(coeff(F[2](x,y,z)/u, v^3));
  Q[2,2](x,y,z) := factor(coeff(F[2](x,y,z), v^4));
```

$$Q_{2,-2}(x, y, z) := (y I - x)^2$$

$$Q_{2,-1}(x, y, z) := 4 (y I - x) z$$

$$Q_{2,0}(x, y, z) := -2 x^2 - 2 y^2 + 4 z^2$$

$$Q_{2,1}(x, y, z) := 4 (x + y I) z$$

$$Q_{2,2}(x, y, z) := (x + y I)^2$$

For $l = 3$, we have

```
> F[3](x,y,z) = (u^2*(x-I*y)-v^2*(x+I*y) - 2*u*v*z)^3;
```

$$F_3(x, y, z) = (u^2 (x - y I) - v^2 (x + y I) - 2 u v z)^3$$

```
> F[3](x,y,z) := collect(expand(rhs(%)), [u,v]);
```

$$F_3(x, y, z) := (-3 I x^2 y + y^3 I + x^3 - 3 x y^2) u^6 + (12 I x y z - 6 x^2 z + 6 y^2 z) v u^5$$
$$+ (3 I x^2 y + 3 I y^3 - 12 I y z^2 - 3 x^3 - 3 x y^2 + 12 x z^2) v^2 u^4 + (12 x^2 z + 12 y^2 z - 8 z^3) v^3 u^3$$
$$+ (3 I x^2 y + 3 I y^3 - 12 I y z^2 + 3 x^3 + 3 x y^2 - 12 x z^2) v^4 u^2$$
$$+ (-12 I x y z - 6 x^2 z + 6 y^2 z) v^5 u + (-3 I x^2 y + y^3 I - x^3 + 3 x y^2) v^6$$

according to which

```
> Q[3,-3](x,y,z) := factor(coeff(F[3](x,y,z), u^6));
  Q[3,-2](x,y,z) := factor(coeff(F[3](x,y,z)/v, u^5));
  Q[3,-1](x,y,z) := factor(coeff(F[3](x,y,z)/v^2, u^4));
  Q[3,0](x,y,z) := factor(coeff(F[3](x,y,z)/v^3, u^3));
  Q[3,1](x,y,z) := factor(coeff(F[3](x,y,z)/u^2, v^4));
  Q[3,2](x,y,z) := factor(coeff(F[3](x,y,z)/u, v^5));
  Q[3,3](x,y,z) := factor(coeff(F[3](x,y,z), v^6));
```

$$Q_{3,-3}(x, y, z) := -(y I - x)^3$$

$$Q_{3,-2}(x, y, z) := -6 z (y I - x)^2$$

$$Q_{3,-1}(x, y, z) := 3 (x^2 + y^2 - 4 z^2) (y I - x)$$

$$Q_{3,0}(x, y, z) := 4 z (3 x^2 + 3 y^2 - 2 z^2)$$

$$Q_{3,1}(x, y, z) := 3 (x^2 + y^2 - 4 z^2) (x + y I)$$

$$Q_{3,2}(x, y, z) := -6 z (x + y I)^2$$

$$Q_{3,3}(x, y, z) := -(x + y I)^3$$

For $l = 4$, we have

```
> F[4](x,y,z) = (u^2*(x-I*y)-v^2*(x+I*y) - 2*u*v*z)^4;
```

$$F_4(x, y, z) = (u^2 (x - y I) - v^2 (x + y I) - 2 u v z)^4$$

```
> F[4](x,y,z) := collect(expand(rhs(%)), [u,v]);
```

$$\begin{aligned} F_4(x, y, z) := & (-4 I x^3 y + 4 I x y^3 + x^4 - 6 x^2 y^2 + y^4) u^8 \\ & + (24 I x^2 y z - 8 I y^3 z - 8 x^3 z + 24 x y^2 z) v u^7 \\ & + (8 I x^3 y + 8 I x y^3 - 48 I x y z^2 - 4 x^4 + 24 x^2 z^2 + 4 y^4 - 24 y^2 z^2) v^2 u^6 \\ & + (-24 I x^2 y z - 24 I y^3 z + 32 I y z^3 + 24 x^3 z + 24 x y^2 z - 32 x z^3) v^3 u^5 \\ & + (6 x^4 + 12 x^2 y^2 - 48 x^2 z^2 + 6 y^4 - 48 y^2 z^2 + 16 z^4) v^4 u^4 \\ & + (-24 I x^2 y z - 24 I y^3 z + 32 I y z^3 - 24 x^3 z - 24 x y^2 z + 32 x z^3) v^5 u^3 \\ & + (-8 I x^3 y - 8 I x y^3 + 48 I x y z^2 - 4 x^4 + 24 x^2 z^2 + 4 y^4 - 24 y^2 z^2) v^6 u^2 \\ & + (24 I x^2 y z - 8 I y^3 z + 8 x^3 z - 24 x y^2 z) v^7 u + (4 I x^3 y - 4 I x y^3 + x^4 - 6 x^2 y^2 + y^4) v^8 \end{aligned}$$

according to which

```
> Q[4,-4](x,y,z) := factor(coeff(F[4](x,y,z), u^8));
  Q[4,-3](x,y,z) := factor(coeff(F[4](x,y,z)/v, u^7));
  Q[4,-2](x,y,z) := factor(coeff(F[4](x,y,z)/v^2, u^6));
  Q[4,-1](x,y,z) := factor(coeff(F[4](x,y,z)/v^3, u^5));
  Q[4,0](x,y,z) := factor(coeff(F[4](x,y,z)/v^4, u^4));
  Q[4,1](x,y,z) := factor(coeff(F[4](x,y,z)/u^3, v^5));
  Q[4,2](x,y,z) := factor(coeff(F[4](x,y,z)/u^2, v^6));
  Q[4,3](x,y,z) := factor(coeff(F[4](x,y,z)/u, v^7));
  Q[4,4](x,y,z) := factor(coeff(F[4](x,y,z), v^8));
```

$$Q_{4,-4}(x, y, z) := (y I - x)^4$$

$$Q_{4,-3}(x, y, z) := 8 z (y I - x)^3$$

$$Q_{4,-2}(x, y, z) := -4 (x^2 + y^2 - 6 z^2) (y I - x)^2$$

$$Q_{4,-1}(x, y, z) := -8 (3 x^2 + 3 y^2 - 4 z^2) (y I - x) z$$

$$Q_{4,0}(x, y, z) := 6 x^4 + 12 x^2 y^2 - 48 x^2 z^2 + 6 y^4 - 48 y^2 z^2 + 16 z^4$$

$$Q_{4,1}(x, y, z) := -8 (3 x^2 + 3 y^2 - 4 z^2) (x + y I) z$$

$$Q_{4,2}(x, y, z) := -4 (x^2 + y^2 - 6 z^2) (x + y I)^2$$

$$Q_{4,3}(x, y, z) := 8 z (x + y I)^3$$

$$Q_{4,4}(x, y, z) := (x + y I)^4$$

Because u and v are arbitrary complex numbers, each $Q_{l,m}$ is a solution of Laplace's equation and, according to the assumptions above, is hence a suitable eigenfunction for the solution of the hydrogen atom in wave mechanics in cartesian coordinates. In general,

```
> -I*(x*Diff(Q[l,m],y) - y*Diff(Q[l,m],x)) = m*Q[l,m];
```

$$-I\left(x\left(\frac{\partial}{\partial y}Q_{l,m}\right)-y\left(\frac{\partial}{\partial x}Q_{l,m}\right)\right)=m\,Q_{l,m}$$

in which m takes values of integers from $-l$ to l. In this representation in terms of $Q_{l,m}$ according to the choice of conditions above, component z of angular momentum, i.e. M_z, has eigenvalue $\frac{m\,h}{2\,\pi}$. The total amplitude function, modified from the result of Fowles and normalized in only the part containing $\sqrt{x^2+y^2+z^2}$, is thus

```
> psi := (k,l,m) ->
  (1/a[mu]/k!*(k+2*l+2)!*(k+2*l+2))^(1/2)/((-1)^l*(2+2*l)!)*Q[l,m](x,y,z)
                     *exp(-sqrt(x^2+y^2+z^2)/(a[mu]*(k+l+1)))

  *LaguerreL(k,2*l+1,(x^2+y^2+z^2)^(1/2)/(a[mu]*(k+l+1)))/(k+l+1)!;
```

$$\psi := (k,l,m)\rightarrow \sqrt{\frac{(k+2\,l+2)!\,(k+2\,l+2)}{a_{\mu}\,k!}}\,Q_{l,m}(x,y,z)\,\mathbf{e}^{\left(-\frac{\sqrt{x^2+y^2+z^2}}{a_{\mu}\,(k+l+1)}\right)}$$
$$\mathrm{LaguerreL}\left(k,2\,l+1,\frac{\sqrt{x^2+y^2+z^2}}{a_{\mu}\,(k+l+1)}\right)\Big/\left((-1)^l\,(2+2\,l)!\,(k+l+1)!\right)$$

We plot a few surfaces, according to the same criteria as in section 12b.53a -- namely at a constant value of ψ equal to $\frac{1}{100}$ of the maximum value for a particular ψ, of amplitude functions directly in these cartesian coordinates according to the above formula, first $\psi_{0,0,0}(x,y,z)$ that is applicable to the hydrogen atom in its ground electronic state. The scale of length in these plots is Bohr radius $a_{\mu}\sim a_0$.

```
> 'psi[0,0,0](x,y,z)' = simplify(psi(0,0,0));
```

$$\psi_{0,0,0}(x,y,z)=\sqrt{\frac{1}{a_{\mu}}}\,\mathbf{e}^{\left(-\frac{\sqrt{x^2+y^2+z^2}}{a_{\mu}}\right)}$$

```
> plot([simplify(eval(psi(0,0,0), [a[mu]=1, y=0,z=0])),
  0.01],x=0..5, -0.1..1.1,
       colour=[red,blue], linestyle=[1,2],
  titlefont=[TIMES,BOLD,14],
       title="amplitude function psi(0,0,0) in direction x",
  titlefont=[TIMES,BOLD,14]);
```




```
>
> plots[implicitplot3d](eval(simplify(psi(0,0,0)),a[mu]=1)=0.01,
    x=-5..5, y=-5..5, z=-5..5, colour=green, grid=[30,30,30],
    axes=boxed, labels=["x","y","z"], title="amplitude function
  psi(0,0,0)",
    scaling=constrained, titlefont=[TIMES,BOLD,14],
  orientation=[140,90]);
```

amplitude function psi(0,0,0)




```
>
```

Next we plot $\psi_{1,0,0}(x, y, z)$.

```
> 'psi[1,0,0](x,y,z)' = simplify(psi(1,0,0));
```

$$\psi_{1,0,0}(x, y, z) = -\frac{3}{8}\,\frac{\sqrt{2}\,\sqrt{\frac{1}{a_\mu}}\,\mathbf{e}^{\left(-1/2\,\frac{\sqrt{x^2+y^2+z^2}}{a_\mu}\right)}\,(-4\,a_\mu + \sqrt{x^2+y^2+z^2}\,)}{a_\mu}$$

```
> plot([simplify(eval(psi(1,0,0), [a[mu]=1, x=0,y=0])), 0.02,
  -0.02], z=-12..12,
       colour=[red, blue, blue], linestyle=[1,2,2],
  titlefont=[TIMES,BOLD,14],
```

```
title="amplitude function psi(1,0,0) in direction z");
```

amplitude function psi(1,0,0) in direction z



\>

This plot of $\psi_{1,0,0}(x, y, z)$ is cut open to reveal the internal structure -- an inner sphere of positive phase surrounded by a spherical shell of negative phase.

```
> plots[implicitplot3d]([eval(simplify(psi(1,0,0)),a[mu]=1)=0.02,

    eval(simplify(psi(1,0,0)),a[mu]=1)=-0.02], x=-0..11,
  y=-11..11,
    z=-11..11, colour=[navy,wheat], grid=[35,35,35],
  scaling=constrained,
    axes=boxed, labels=["x","y","z"], title="amplitude function
  psi(1,0,0)",
```

```
titlefont=[TIMES,BOLD,14], orientation=[134,90]);
```

amplitude function psi(1,0,0)




```
>
```

We next plot amplitude functions with $l = 1$; first $\psi_{0,\,1,\,0}(x, y, z)$ that has only a real part,

```
> 'psi[0,1,0](x,y,z)' = simplify(psi(0,1,0));
```

$$\psi_{0,\,1,\,0}(x, y, z) = \frac{1}{6}\sqrt{6}\,\sqrt{\frac{1}{a_{\mu}}}\; z\, \mathbf{e}^{\left(-1/2\,\frac{\sqrt{x^2+y^2+z^2}}{a_{\mu}}\right)}$$

```
> plot([simplify(eval(psi(0,1,0), [a[mu]=1, x=0,y=0])), 0.006,
  -0.006], z=-15..15,
       colour=[red, blue, blue], linestyle=[1,2,2],
  titlefont=[TIMES,BOLD,14],
```

```
title="amplitude function psi(0,1,0) in direction z");
```




```
>
> plots[implicitplot3d]([eval(simplify(psi(0,1,0)),a[mu]=1)=0.006
,
   eval(simplify(psi(0,1,0)),a[mu]=1)=-0.006], x=-12.5..12.5,
y=-12.5..12.5,
   z=-14..14, colour=[red,blue], grid=[30,30,30],
scaling=constrained,
   axes=boxed, labels=["x","y","z"], title="amplitude function
psi(0,1,0)",
   titlefont=[TIMES,BOLD,14], orientation=[0,90],
colour=[red,blue]);
```

amplitude function psi(0,1,0)



\>

and then $\psi_{0,1,1}(x, y, z)$ that has both real and imaginary parts that we plot separately.

```
> 'psi[0,1,1](x,y,z)' = simplify(psi(0,1,1));
```

$$\psi_{0,1,1}(x, y, z) = \frac{1}{12}\sqrt{6}\sqrt{\frac{1}{a_\mu}}\,(x + y\,I)\,\mathbf{e}^{\left(-1/2\,\frac{\sqrt{x^2+y^2+z^2}}{a_\mu}\right)}$$

```
> plot([simplify(eval(Re(psi(0,1,1)), [a[mu]=1, y=0,z=0])), 
  0.003, -0.003], x=-15..15,
       colour=[red, blue, blue], linestyle=[1,2,2], 
  titlefont=[TIMES,BOLD,14],
       title="amplitude function psi(0,1,1) in direction x");
```




```
>
> plots[implicitplot3d]([eval(simplify(Re(psi(0,1,1))),a[mu]=1)=0
.003,
   eval(simplify(Re(psi(0,1,1))),a[mu]=1)=-0.003], x=-14..14,
y=-11..11,
   z=-11..11, colour=[red,blue], grid=[30,30,30],
scaling=constrained,
   axes=boxed, labels=["x","y","z"], title="real part of
amplitude function psi(0,1,1)",
   titlefont=[TIMES,BOLD,14], orientation=[-90,90],
colour=[coral,violet]);
```

real part of amplitude function psi(0,1,1)




```
>
> plots[implicitplot3d]([eval(simplify(Im(psi(0,1,1))),a[mu]=1)=0
  .003,
     eval(simplify(Im(psi(0,1,1))),a[mu]=1)=-0.003], x=-11..11,
  y=-14..14,
     z=-11..11, colour=[red,blue], grid=[30,30,30],
  scaling=constrained,
     axes=boxed, labels=["x","y","z"], title="imaginary part of
  amplitude function psi(0,1,1)",
     titlefont=[TIMES,BOLD,14], orientation=[0,90],
  colour=[maroon,wheat]);
```

imaginary part of amplitude function psi(0,1,1)



>

The remaining plots illustrate amplitude functions with $l = 2, 3, 4$.

```
> 'psi[0,2,0](x,y,z)' = simplify(psi(0,2,0));
```

$$\psi_{0,2,0}(x, y, z) = -\frac{1}{180}\sqrt{30}\sqrt{\frac{1}{a_\mu}}\,(x^2 + y^2 - 2z^2)\,\mathbf{e}^{\left(-1/3\frac{\sqrt{x^2+y^2+z^2}}{a_\mu}\right)}$$

```
> plot([simplify(eval(psi(0,2,0), [a[mu]=1, x=0,y=0])), 0.003],
  z=-32..32,
       colour=[red, blue, blue], linestyle=[1,2,2],
  titlefont=[TIMES,BOLD,14],
       title="amplitude function psi(0,2,0) in direction z");
```

amplitude function psi(0,2,0) in direction z




```
>
> plots[implicitplot3d]([eval(simplify(psi(0,2,0)),a[mu]=1)=0.003
  ,
     eval(simplify(psi(0,2,0)),a[mu]=1)=-0.012], x=-23..23,
  y=-23..23,
     z=-30..30, colour=[red,blue], grid=[30,30,30],
  scaling=constrained,
     axes=boxed, labels=["x","y","z"], title="amplitude function
  psi(0,2,0)",
     titlefont=[TIMES,BOLD,14], orientation=[0,60],
  colour=[pink,cyan]);
```

amplitude function psi(0,2,0)




```
>
> 'psi[0,3,0](x,y,z)' = simplify(psi(0,3,0));
```

$$\psi_{0,3,0}(x,y,z) = -\frac{1}{840}\,\mathbf{e}^{\left(-1/4\,\frac{\sqrt{x^2+y^2+z^2}}{a_\mu}\right)}\sqrt{\frac{1}{a_\mu}}\left(x^2+y^2-\frac{2\,z^2}{3}\right)z\sqrt{35}$$

```
> plot([simplify(eval(psi(0,3,0), [a[mu]=1, x=0,y=0])), 0.004,
  -0.004], z=-52..52,
       colour=[red, blue, blue], linestyle=[1,2,2],
  titlefont=[TIMES,BOLD,14],
       title="amplitude function psi(0,3,0) in direction z");
```




```
>
> plots[implicitplot3d]([eval(simplify(psi(0,3,0)),a[mu]=1)=0.004
  ,
     eval(simplify(psi(0,3,0)),a[mu]=1)=-0.004], x=-40..0,
  y=-40..40,
     z=-47..47, colour=[magenta,coral], grid=[25,25,25],
  scaling=constrained,
     axes=boxed, labels=["x","y","z"], title="amplitude function
  psi(0,3,0)",
     titlefont=[TIMES,BOLD,14], orientation=[0,60]);
```

amplitude function psi(0,3,0)




```
>
> 'psi[0,4,0](x,y,z)' = simplify(psi(0,4,0));
```

$$\psi_{0,4,0}(x,y,z) =$$

$$\frac{1}{100800}\,\mathbf{e}^{\left(-1/5\,\frac{\sqrt{x^2+y^2+z^2}}{a_\mu}\right)}\sqrt{\frac{1}{a_\mu}}\left(x^4+(2\,y^2-8\,z^2)\,x^2+y^4-8\,y^2\,z^2+\frac{8\,z^4}{3}\right)\sqrt{70}$$

```
> plot([simplify(eval(psi(0,4,0), [a[mu]=1, x=0,y=0])), 0.0065],
  z=-72..72,
       colour=[red, blue, blue], linestyle=[1,2,2],
  titlefont=[TIMES,BOLD,14],
       title="amplitude function psi(0,3,0) in direction z");
```

amplitude function psi(0,3,0) in direction z




```
>
> plots[implicitplot3d]([eval(simplify(psi(0,4,0)),a[mu]=1)=0.006
  5,
     eval(simplify(psi(0,4,0)),a[mu]=1)=-0.0065], x=-62..0,
  y=-62..62,
     z=-68..68, colour=[maroon,yellow], grid=[25,25,25],
  scaling=constrained,
     axes=boxed, labels=["x","y","z"], title="amplitude function
  psi(0,4,0)",
     titlefont=[TIMES,BOLD,14], orientation=[0,60]);
```

amplitude function psi(0,4,0)



>

The surfaces of other amplitude functions or their real and imaginary parts can be analogously plotted.

This derivation, following Fowles, includes some assumptions and questionable arguments, for instance, the assumption that the terms in x,y,z, in $\mathrm{F}(x, y, z)$ satisfy Laplace's equation, which in general has multiple solutions, cf section 7.41, although the method was clearly designed to yield results analogous to those in the system of spherical polar coordinates. The solution above is a partial solution of a particular type; for a true separation the basis eigenfunctions have a form $X_n(x)\, Y_n(y)\, Z_n(z)$. The partial separation implemented above involves two variables $a\,x + b\,y + c\,z$ and r, i.e. $F_l(a\,x + b\,y + c\,z)\,\mathrm{R}(r)$; these variables depend on three parameters a, b, c; a complete set requires varying these parameters. The resulting eigenfunctions are not orthogonal; for each choice of parameters the eigenfunction are linear combinations of separable solutions in spherical polar coordinates. The plots above show a satisfactory correlation with the plots for the amplitude functions in spherical polar coordinates in section 12b.53b. The results are hence of interest in the context of multiple coordinates systems for the hydrogen atom according to Schroedinger's equations. Another choice of solution of Laplace's equation might yield amplitude functions in cartesian coordinates that simulate amplitude functions as solutions of Schroedinger's temporally independent equation in paraboloidal, ellipsoidal or spheroconical coordinates.

>